Constrained minimization

Minimize (x=i, y=R)

```
> SUBS := i=x, R=y ;
```

$$SUBS := i = x,\ R = y \tag{1}$$

```
> f := -subs(SUBS,R*i^2) ;
```

Maximize the power

$$f := -y\,x^2 \tag{2}$$

Constraints

```
> h := V-subs(SUBS,(R+Ri)*i) ;
  g := y >= 0 ;
```

$$h := V - (y + Ri)\,x$$

$$g := 0 \le y \tag{3}$$

```
> L := f-lambda*h -mu*rhs(g) ;
```

$$L := -y\,x^2 - \lambda\,(V - (y + Ri)\,x) - \mu\,y \tag{4}$$

Nonlinear system

```
> EQ1 := diff(L,x) ;
  EQ2 := diff(L,y) ;
  EQ3 := h ;
  EQ4 := rhs(g)*mu ;
```

$$EQ1 := -2\,x\,y - \lambda\,(-Ri - y)$$

$$EQ2 := \lambda\,x - x^2 - \mu$$

$$EQ3 := V - (y + Ri)\,x$$

$$EQ4 := \mu\,y \tag{5}$$

```
> SOL := solve( {EQ||(1..4)}, {x,y,lambda,mu} ) ;
```

$$SOL := \left\{\lambda = \frac{1}{2}\,\frac{V}{Ri},\ \mu = 0,\ x = \frac{1}{2}\,\frac{V}{Ri},\ y = Ri\right\},\ \left\{\lambda = 0,\ \mu = -\frac{V^2}{Ri^2},\ x = \frac{V}{Ri},\ y = 0\right\} \tag{6}$$

Check if the solution is a mimimum. Compute the gradient of the active constraints

```
> subs(SOL,rhs(g)) ; # g is not active
```

$$Ri \tag{7}$$

```
> G := subs(SOL[1],<diff(h,x)|diff(h,y)>) ;
```

$$G := \begin{bmatrix} -2\,Ri & -\frac{1}{2}\,\frac{V}{Ri} \end{bmatrix} \tag{8}$$

```
> K := op(LinearAlgebra[NullSpace](G)) ;
```

$$K := \begin{bmatrix} -\frac{1}{4}\,\frac{V}{Ri^2} \\ 1 \end{bmatrix} \tag{9}$$

The hessian of the Lagrangian

```
> H := <<diff(L,x,x),diff(L,x,y)>|
        <diff(L,x,y),diff(L,y,y)>>;
  H := subs(SOL[1],H) ;
```

$$H := \begin{bmatrix} -2\,y & -2\,x+\lambda \\ -2\,x+\lambda & 0 \end{bmatrix}$$

$$H := \begin{bmatrix} -2\,Ri & -\frac{1}{2}\,\frac{V}{Ri} \\ -\frac{1}{2}\,\frac{V}{Ri} & 0 \end{bmatrix} \quad \textbf{(10)}$$

```
> LinearAlgebra[Transpose](K).H.K ;
```

$$\frac{1}{8}\,\frac{V^2}{Ri^3} \quad \textbf{(11)}$$